
13 Step-Path Failure of Rock Slopes

13.1 Problem Statement

While UDEC represents a jointed rock structure as a system of discrete blocks by default, it is also
a straightforward matter to simulate the case in which joints terminate within intact rock. This is
illustrated in this example application, which analyzes the stability of a simple slope containing
noncontinuous en echelon jointing. If failure occurs both along the joints and within the intact rock,
a “step-path failure” involving a combination of shear failure along the joints and shear and tensile
failure within the intact rock bridging between the joints can develop.

Three en echelon joints exist within the slope, as depicted in Figure 13.1. The three joints have a
dip of 35◦, and terminate within the rock near the slope surface. The slope dips at 50◦, and has a
height of 11.8 m.

Figure 13.1 Geometry of simple slope containing en echelon joints

The properties of the intact rock are

density 2000 kg/m3

Young’s modulus 20 GPa
Poisson’s ratio 0.3
friction angle 25◦
dilation angle 0
cohesion 25,000 Pa
tensile strength 0

The properties of the en echelon joints are

normal stiffness 100 GPa/m
shear stiffness 10 GPa/m
friction angle 35◦
cohesion 1000 MPa

The failure mode within the slope model is evaluated by performing factor-of-safety calculations
based upon the shear strength reduction method (see Section 2 in Theory and Background). The
failure mode can be identified when the strength properties are reduced to produce the state at which
failure occurs.

13.2 UDEC Analysis

Discontinuous jointing can be simulated in UDEC by combining real joints with construction joints
along the paths of the en echelon joints. Note that construction joints effectively glue two blocks
together so that the blocks behave as one intact block. The UDEC commands to create the en
echelon joints are described below, and the resulting joint structure is illustrated in Figure 13.2.

Cracks are introduced along the path of the three en echelon joints by using the block cut crack
command to create real joints, and the block cut crack ... join command to create a construction
joint connecting the ends of real joints. Several commands are executed:

block cut crack (17,8.2) (18.33,9.37) join
block cut crack (18.33,9.37) (22.33,12.25)
block cut crack (22.33,12.25) (22.33,13.5) join
block cut crack (22.33,13.5) (26.26,16)
block cut crack (26.26,16) (26.26,17.16) join
block cut crack (26.26,17.16) (29.33,19.33)
block cut crack (29.33,19.33) (30,20) join

Each block cut crack ... join command creates a construction joint that connects the ends of the real
joints to produce a continuous joint through the model. Figure 13.2 displays the real joints, but not
the construction joints.

For comparison, two separate models are created: the first contains no joints, and the second is
created with three continuous joints located at the same position and orientation as the en echelon
joints. The three continuous joints are created with the commands

block cut crack (17,8.2) (33.2,20)
block cut crack (15.5,8.2) (31.7,20)
block cut crack (14,8.2) (30.2,20)

This joint structure is shown in Figure 13.3.

Figure 13.2 En echelon joints in slope model

Figure 13.3 Continuous joints in slope model

A factor-of-safety calculation (block factory-of-safety) is performed for all three models. The factor
of safety for the en echelon joint model is found to be 1.28. A step-path failure mode is indicated
by the velocity vectors plot shown in Figure 13.4, and by the contour plot of joint maximum shear
displacement and rock maximum shear strain shown in Figure 13.5.

For comparison, the factor of safety was also calculated for the model with no joints (FOS = 1.38,
see Figure 13.6) and for the model with continuous joints through the slope (FOS = 1.01, see
Figure 13.7). A circular failure mode within the intact rock is indicated in Figure 13.6. A planar
shear failure mode along the joints is shown in Figure 13.7.

Figure 13.4 Step-path failure of slope – factor-of-safety = 1.28

Figure 13.5 Step-path failure depicted by joint maximum shear displacement
and rock maximum shear strain

UDEC Version 7.0



13 - 6 Example Applications

Figure 13.6 Continuum failure of slope – factor-of-safety = 1.38

Figure 13.7 Continuous joint shear failure of slope – factor-of-safety = 1.01

13.3 Listing of Data File

Example 13.1 STEPPATH.DAT

;File:steppath.dat
model Title ’Step-Path Slope Failure’
model new
; model with en-echelon joint structure
block tolerance corner-round-length 4.5E-2
block tolerance minimum-edge-length 9E-2
block create polygon 0 0 0 20 45 20 45 0
; create en-echelon joints
block cut crack (0,8.2) (17,8.2)
block cut crack (17,8.2) (26.9,20)
block cut crack (17,8.2) (18.33,9.37) join
block cut crack (18.33,9.37) (22.33,12.25)
block cut crack (22.33,12.25) (22.33,13.5) join
block cut crack (22.33,13.5) (26.26,16)
block cut crack (26.26,16) (26.26,17.16) join
block cut crack (26.26,17.16) (29.33,19.33)
block cut crack (29.33,19.33) (30,20) join
block delete range pos-x 0,17 pos-y 8,20
block zone gen edge 1.0
model save ’sp1.sav’
; assign properties
block zone group ’rock’
block zone cmodel assign mohr-c density 2E3 bulk 1.66667E10 ...

shear 7.69231E9 friction 25 cohesion 2.5E4 range group ’rock’
block contact property material 1 stiffness-shear 1E10 ...

stiffness-normal 1E11 friction 35 cohesion 1E3
block contact group ’joint’
block contact cmodel assign area st-s 1E10 st-n 1E11 friction 35 ...

cohesion 1E3 range group ’joint’
; new contact default
block contact cmodel default area st-s=1E10 st-n=1E11 friction=35 ...

cohesion=1000
; set boundary conditions
block gridpoint apply velocity-x 0 range pos-x -0.1 0.1 pos-y -0.1 8.2
block gridpoint apply velocity-x 0 range pos-x 44.9 45.1 pos-y -0.1 20.1
block gridpoint apply velocity-y 0 range pos-x -0.1 45.1 pos-y -0.1 0.1
model gravity 0.0 -9.81
model save ’sp2.sav’
block solve ratio 1.0E-5 elastic
model save ’sp3.sav’
; Factor of safety
block factor-of-safety no_restore file ’FoSmode1.fsv’

model save ’FoSmode1.fsv’
;
;
model new
; model with continuous joint structure
block tolerance corner-round-length 4.5E-2
block tolerance minimum-edge-length 9E-2
block create polygon 0 0 0 20 45 20 45 0
; create continuous joints
block cut crack 0 8.2 17 8.2
block cut crack 17 8.2 26.9 20
block cut crack 17 8.2 33.2 20
block cut crack 15.5 8.2 31.7 20
block cut crack 14 8.2 30.2 20
block delete range pos-x 0 19 pos-y 8 20
block zone gen edge 1.0
model save ’sp1a.sav’
; assign properties
block zone group ’rock’
block zone cmodel assign mohr-c density 2E3 bulk 1.66667E10 ...

shear 7.69231E9 friction 25 cohesion 2.5E4 range group ’rock’
block contact group ’joint’
block contact cmodel assign area st-s 1E10 st-n 1E11 friction 35 ...

cohesion 1E3 range group ’joint’
; new contact default
block contact cmodel default area st-s 1E10 st-n 1E11 friction 35 ...

cohesion 1000
; set boundary conditions
block gridpoint apply velocity-x 0 range pos-x -0.1 0.1 pos-y -0.1 8.2
block gridpoint apply velocity-x 0 range pos-x 44.9 45.1 pos-y -0.1 20.1
block gridpoint apply velocity-y 0 range pos-x -0.1 45.1 pos-y -0.1 0.1
model gravity 0.0 -9.81
model save ’sp2a.sav’
block solve ratio 1.0E-5 elastic
model save ’sp3a.sav’
; Factor of safety
block factor-of-safety no_restore file ’FoSmode2.fsv’
model save ’FoSmode2.fsv’
;
model new
; model with no joints
block tolerance corner-round-length 4.5E-2
block tolerance minimum-edge-length 9E-2
block create polygon 0 0 0 8.2 17 8.2 26.9 20 45 20 45 0
block zone gen edge 1.0
model save ’sp1b.sav’

; assign properties
block zone group ’rock’
block zone cmodel assign mohr-c density 2E3 bulk 1.66667E10 ...

shear 7.69231E9 friction 25 cohesion 2.5E4 range group ’rock’
; set boundary conditions
block gridpoint apply velocity-x 0 range pos-x -0.1 0.1 pos-y -0.1 8.2
block gridpoint apply velocity-x 0 range pos-x 44.9 45.1 pos-y -0.1 20.1
block gridpoint apply velocity-y 0 range pos-x -0.1 45.1 pos-y -0.1 0.1
model gravity 0.0 -9.81
model save ’sp2b.sav’
block solve ratio 1.0E-5 elastic
model save ’sp3b.sav’
; Factor of safety
block factor-of-safety no_restore file ’FoSmode3.fsv’
model save ’FoSmode3.fsv’
ret

